802 Project Summary

Maksuda Aktar Toma, Jo Charbonneau, Ryan Lalicker

December 9, 2024

Introduction

This paper summarizes the consulting that was done for our assigned STAT 802 group. For more information on the experiment, the data, or any other files used in this paper see our Github page which can be found at <https://github.com/maksudatoma/Stat-802-Project>. The coding languages used in the paper are R and SAS. The corresponding code can be found in *Appendix A - R Code* and *Appendix B - SAS Code* respectively.

Initial Meetings

The first meeting with our clients was on September 13th. We discussed their project and what kind of data they were going to be looking at. They detailed to us their project, which is looking at the levels of Salmonella in beef jerky at different inoculations and thicknesses. Prior to the meeting they sent us what their variables would be, which gave us a good idea of what might be the best experimental design. The group informed us they were avoiding a completely randomized design (CRD) at the request of their professor. With that in mind, we suggested other possible models.

Later, after receiving feedback from Dr. Howard and several PhD students within the statistics department, we suggested adding a time component to the experiment as well as creating multiple batches to replicate each treatment combination. This lead to us suggested a mixed model for the analysis approach.

In both the initial meeting and the follow-up session the clients were more than happy to implement our suggestions. In the end the experiment involved two thickness levels (one-fourth and one-eighth of an inch), two inoculation methods (dry and wet), and five evenly spaced time points were measurements were taken (weeks 1-5) creating twenty entries per batch. The exact number of batches would not be known until after the power analysis, found in the *Power Analysis* section. We provided the client an example dataset we created to give them a better idea of what the end product may look like. This dataset had five batches.

Study Objectives and Proposed Model

The clients were most interested in the effect of the thickness levels, the inoculation method, and their interaction had on the Salmonella levels. In the final model we included the week effect and subsequent interactions as well. These variables are the fixed effects in the proposed mixed model.

The other variable included in the experiment is the batch number. This is therefore treated as a random variable. As mentioned above, the exact number of batches needed was unknown prior to the power analysis, but five was used as a starting value.

Overall the study employs a 2×2 factorial design with two main factors: Inoculation Method (Dry, Wet) and Thickness (1/4-inch, 1/8-inch). Repeated measurements are taken over five equally spaced time points (Weeks 1 to 5), allowing the analysis of both main effects, their interaction, and changes over time.

The model can be written in the form

$$Y_{ijkl} = \mu + \alpha_i + \beta_j + \tau_k + (\alpha\beta)_{ij} + (\alpha\tau)_{ik} + (\beta\tau)_{jk} + (\alpha\beta\tau)_{ijk} + u_l + e_{ijkl}$$

Here, Y_{ijkl} is the Salmonella level and μ is the overall mean. The fixed effects are represented by α_i for the effect of the ith inoculation method, β_j for the effect of the jth thickness level, and τ_k for the effect of the kth week. The interaction effect of the ith inoculation method and the jth thickness level is represented by $(\alpha\beta)_{ij}$, with the other two-way interactions following this form. The three-way interaction between all fixed effects is represented as $(\alpha\beta\tau)_{ijk}$. The random effect for batches is represented by u_l, which we assume are distributed as $u_l \sim N(0, \sigma_u^2)$. Lastly, the residuals are represented by e_{ijkl}, which we assume can be distributed as $e_{ijkl} \sim N(0, \sigma^2)$.

Power Analysis

Before power analysis, We reached out to the client later on in the process to determine what contrasts they were most interested in testing. They expressed they wanted to see the difference between the two levels of the inoculation method, the two levels of the thickness, and the orthogonal contrasts these variable. This resulted in six contrasts being tested.

To determine the necessary number of batches needed to increase the likelihood of detecting a true treatment effect, we performed a power analysis. To do this, probable treatment mean estimates across all five weeks and variance estimates were needed. The clients provided these metrics from Brown et al. (2024). We then used these metrics to create a dataset with five batches where the response variable was identical across the batches. This dataset was then evaluated to determine the power. The results of the power analysis perfomed in SAS are shown below.

Obs	Label	NumDF	DenDF	FValue	ProbF	Effect	ncparm	alpha	fcrit	power
1	Dry vs Wet	1	76	9.23	0.0033		9.226	0.05	3.96676	0.85058
2	1/4 vs 1/8 inches	1	76	103.28	<.0001		103.276	0.05	3.96676	1.00000
3	Dry vs Wet at 1/4 Inches	1	76	5.70	0.0195		5.695	0.05	3.96676	0.65406
4	Dry vs Wet at 1/8 Inches	1	76	44.65	<.0001		44.651	0.05	3.96676	1.00000
5	1/4 vs 1/8 inches for Dry inoculation	1	76	137.36	<.0001		137.364	0.05	3.96676	1.00000
6	1/4 vs 1/8 inches for Wet inoculation	1	76	7.03	0.0097		7.031	0.05	3.96676	0.74475
7		1	76	9.23	0.0033	Inoculation_Method	9.226	0.05	3.96676	0.85058
8		1	76	103.28	<.0001	Thickness	103.276	0.05	3.96676	1.00000
9		1	76	41.12	<.0001	Inoculatio*Thickness	41.120	0.05	3.96676	0.99999
10		4	76	7.82	<.0001	Week	31.280	0.05	2.49205	0.99659
11		4	76	13.77	<.0001	Inoculation_Met*Week	55.092	0.05	2.49205	1.00000
12		4	76	20.78	<.0001	Thickness*Week	83.132	0.05	2.49205	1.00000
13		4	76	4.04	0.0050	Inocula*Thickne*Week	16.162	0.05	2.49205	0.89525

Figure 1: Results of power analysis.

The first six rows of the table correspond to the contrasts the clients were interested in testing, while the bottom seven rows are measuring the fixed effects of the model. Many of the terms have more than 80% power. Specifically the fixed effects were all high enough for both the clients and ourselves to feel comfortable using five batches. Two of the orthogonal contrasts, `Dry vs Wet at 1/4 Inches` and `1/4 vs 1/8 inches for Wet inoculation` did have lower power scores, but after talking with both the clients and Dr. Howard about them, we felt comfortable to proceed.

Simulating Data

After finding the necessary number of batches, which was five, we proceeded with simulating the data. The estimated treatment means and variances provided by the client were used in the simulation as well. We then reviewed the simulated dataset for major issues, such as negative response values, and reran the power analysis on the new data set to ensure everything was working properly. After finding no problems with the dataset, we sent it to the clients. Note, the simulation was performed in SAS.

Data Analysis

Summary Statistics

As part of the project, we analyzed the simulated dataset. Before fitting out model to the dataset, we first wanted to explore some of the variables. Figure 2 shows the mean values and

standard deviations for each treatment combination. We can see the changes in mean values are small, so further exploration and analysis are needed.

Inoculation Method	Thickness	Week	Mean Response	SD Response	Count
Dry	1/4-inch	1	4.199	0.187	5
Dry	1/4-inch	2	4.137	0.163	5
Dry	1/4-inch	3	4.187	0.294	5
Dry	1/4-inch	4	4.306	0.272	5
Dry	1/4-inch	5	4.645	0.178	5
Dry	1/8-inch	1	4.801	0.173	5
Dry	1/8-inch	2	4.801	0.238	5
Dry	1/8-inch	3	4.596	0.235	5
Dry	1/8-inch	4	4.425	0.206	5
Dry	1/8-inch	5	4.876	0.177	5
Wet	1/4-inch	1	4.240	0.307	5
Wet	1/4-inch	2	4.523	0.236	5
Wet	1/4-inch	3	4.534	0.173	5
Wet	1/4-inch	4	4.357	0.421	5
Wet	1/4-inch	5	4.299	0.200	5
Wet	1/8-inch	1	4.697	0.181	5
Wet	1/8-inch	2	4.809	0.327	5
Wet	1/8-inch	3	4.530	0.215	5
Wet	1/8-inch	4	4.223	0.361	5
Wet	1/8-inch	5	4.151	0.295	5

Figure 2: Summary of response rariable across factors and weeks

Distribution of response variable

Before continuing our investigation into the relationships among the treatment variables, we want to look into the response variable (Salmonella levels). Specifically, we want to see how it is distributed. Figure 3 shows a histogram and Q-Q plot of the response variable in the left and right plots respectively. While the histogram shows a slight potential skew, this is not enough for us say the distribution is non-normal. Furthermore, the Q-Q plot indicates the response variable follows a relatively normal distribution.



(a) Histogram (b) Q-Q plot

Figure 3: Plots to see the distribution of the response variable.

Exploring the Data

Now we will graphically look at how the different variables of the model impact the response variable. Figure 4 shows how the mean response variables we saw in Figure 2 change over time, while also controlling for thickness and inoculation method. Note the y-axis of plot does not start at the origin. We can see the mean values of samples that used a wet inoculation method (blue lines) tended decrease over time, while samples with the dry inoculation method (red lines) were more of a mixed bag but saw sharp increases between weeks four and five. The samples cut to 1/8 inches thick (dashed lines) were very similar for most weeks, but diverged near the end of the experiment, while samples cut to 1/4 inches thick (solid lines) did not seem as similar.



Figure 4: Line plot of response variable over time, controlling for thickness and inoculation method.

Another element to consider is correlation over time. Since this is a repeated measures experiment we need to account for this correlation by selecting a type of covariance structure. There are several types of structures such as variance components (VC), unstructured (UN), compound symmetry (CS) and heterogeneous compound symmetry (CSH), p-order auto-regressive (AR(p)) and heterogeneous auto-regressive (ARH(p)), p-order ante-dependence (ANTE(p)), and Toeplitz (TOEP) among others. For more information on these see Lipka and Tyner (2004) and USDA (n.d.). The AR(p) structure fits data that is ordered through time an equally spaced. For that reason, our initial plan was to use this as the covariance structure with $p = 1$.

To see if this first-order auto-regressive structure might fit the data, let's consider the table and plot in Figure 5. The correlation matrix (left) and plot (right) show the relationships between repeated measurements over weeks one to five. Strong correlations are observed between adjacent weeks (e.g., Week 1 vs. Week 2, r=0.69, Week 2 vs. Week 3, r=0.74), indicating temporal dependency. Correlations weaken as the time gap increases (e.g., Week 1 vs. Week 5, r=0.29), leading us to believe orders of $p > 1$ are not necessary. This can be seen visually in the plot which uses circle size and color to model the correlation metrics seen in the matrix.

We can see as the gap between weeks increases, the circles become smaller and lighter. This pattern supports the use of models like AR(1).

Time	1	2	3	4	5
1	1.00	0.69	0.55	0.21	0.29
2	0.69	1.00	0.74	0.50	0.16
3	0.55	0.74	1.00	0.52	0.17
4	0.21	0.50	0.52	1.00	0.45
5	0.29	0.16	0.17	0.45	1.00

(a) Correlation matrix



(b) Correlation matrix

Figure 5: Table and plots to see the correlation across time.

Model Comparison

After exploring the data we can move on to fitting the model. While we were confident in using the AR(1), we chose to fit the model using other covariance structures as well so we could see how the fit compares. The results are shown in Figure 6. For each of the fit statistics in this table a lower score is better, even when looking at negative values. (Bobbit (2021)). This means the model fit using an AR(1) structure had the best AIC and AICC scores and a respectable BIC score. This verifies our choice in the AR(1) covariance structure.

Model	AIC	AICC	BIC
VC	-12.25	-12.09	-13.03
UN	-12.22	-3.58	-18.47
CS	-11.60	-11.29	-12.77
AR(1)	-12.86	-12.55	-14.03
ARH(1)	-12.56	-11.01	-15.30
ANTE(1)	-8.57	-5.39	-12.48
TOEP	-7.34	-6.19	-9.68

Figure 6: Model comparison table.

After verifying the fit of the AR(1) covariance structure for the repeated measures, we needed to see if the assumptions for a linear mixed model were violated or not. These include the residuals being normally distributed and homogeneous. The plots below allow us to evaluate

these assumptions. To graphically test normality, we can look at both the histogram (top right) and the Q-Q plot (bottom left). These both appear approximately normal, indicating the assumption holds. The boxplot (bottom right) can also show normality as well as potential outliers. It appears there is one outlier, but the normallity assmption still holds. The residual plot (top left) allows us to check if the homogeneous assumption holds, and it appears to since the points seem somewhat randomly distributed with no clear pattern. Since the assumptions are holding, we can proceed with the linear mixed model.



Figure 7: Residual plots for checking assumptions.

Model Output

I dont't think we need this - Ryan PROC GLIMMIX was implemented in SAS where reponse variable (y) is Gaussian and the link function used is 'identity'. The 'class level information' table displays the number of levels and their values in the dataset for categorical variables. The model successfully converges. Various model fit statistics are produced.

The GLIMMIX Procedure

Model Information	
Data Set	WORK.DATA
Response Variable	Response
Response Distribution	Gaussian
Link Function	Identity
Variance Function	Default
Variance Matrix Blocked By	Batches
Estimation Technique	Restricted Maximum Likelihood
Degrees of Freedom Method	Containment

Class Level Information		
Class	Levels	Values
Batches	5	1 2 3 4 5
Inoculation_Method	2	Dry Wet
Thickness	2	1/4-inch 1/8-inch
Week	5	1 2 3 4 5

Number of Observations Read	100
Number of Observations Used	100

Figure 8: Fig

Below are three tables from the SAS output. The first is the *Fit Statistics* table, which is where the metrics used in Figure 6 come from. The *Covariance Parameter Estimates* table shows how much of the variance in the model is explained by the random terms. The *Type III Tests of Fixed Effects* table allows us to see if the fixed effects and/or their interactions are significant by looking at the p-values reported in the `Pr > F` column.

Fit Statistics	
-2 Res Log Likelihood	-18.86
AIC (smaller is better)	-12.86
AICC (smaller is better)	-12.55
BIC (smaller is better)	-14.03
CAIC (smaller is better)	-11.03
HQIC (smaller is better)	-16.01
Generalized Chi-Square	2.13
Gener. Chi-Square / DF	0.03

Covariance Parameter Estimates			
Cov Parm	Subject	Estimate	Standard Error
Intercept	Batches	0.03802	0.02755
AR(1)	Batche*Inocul*Thickn	-0.2140	0.1286
Residual		0.02664	0.004423

Type III Tests of Fixed Effects				
Effect	Num DF	Den DF	F Value	Pr > F
Inoculation_Method	1	76	4.98	0.0286
Thickness	1	76	81.92	<.0001
Inoculatio*Thickness	1	76	32.65	<.0001
Week	4	76	6.12	0.0003
Inoculation_Met*Week	4	76	17.90	<.0001
Thickness*Week	4	76	14.59	<.0001
Inocula*Thickne*Week	4	76	0.55	0.6970

Figure 9: *Fit Statistics*, *Covariance Parameter Estimates*, and *Type III Tests of Fixed Effects* tables.

The estimated variance for the random intercept (batches) is 0.03802, with a standard error of 0.02755. The estimate for the AR(1) autocorrelation structure is -0.2140, capturing the weak negative correlation between adjacent time points. Lastly, the model fits the data reasonably well, with the residual variance (0.02064) indicating minimal unexplained variability, and the SE (0.004423) confirming the reliability of this estimate.

For the fixed effects we first need to consider the interaction terms. While the three-way interaction between the inoculation method, thickness, and week is not significant, all three two-way interactions between these variables are highly significant ($p<0.0001$). Therefore, we need to consider simple effects. Overall we can say each respective two-way interaction does

have an impact on the data.

The next table shows the least squares means (LSMeans) for combinations of the inoculation method, thickness, and week, representing the estimated response for each factor combination. The estimates are highly significant ($p<0.0001$) for all combinations, with confidence intervals providing ranges for each mean. Key statistics include the standard error, degrees of freedom, t-values, and p-values, reflecting the precision and significance of the estimates.

Inocula*Thickne*Week Least Squares Means

Inoculation_Method	Thickness	Week	Estimate	Standard Error	DF	t Value	Pr > |t|	Alpha	Lower	Upper
Dry	1/4-inch	1	4.1992	0.1137	76	36.93	<.0001	0.05	3.9727	4.4257
Dry	1/4-inch	2	4.1373	0.1137	76	36.38	<.0001	0.05	3.9108	4.3637
Dry	1/4-inch	3	4.1866	0.1137	76	36.82	<.0001	0.05	3.9601	4.4131
Dry	1/4-inch	4	4.3063	0.1137	76	37.87	<.0001	0.05	4.0798	4.5328
Dry	1/4-inch	5	4.6451	0.1137	76	40.85	<.0001	0.05	4.4186	4.8716
Dry	1/8-inch	1	4.8012	0.1137	76	42.22	<.0001	0.05	4.5747	5.0277
Dry	1/8-inch	2	4.8007	0.1137	76	42.21	<.0001	0.05	4.5742	5.0272
Dry	1/8-inch	3	4.5957	0.1137	76	40.41	<.0001	0.05	4.3692	4.8222
Dry	1/8-inch	4	4.4247	0.1137	76	38.91	<.0001	0.05	4.1982	4.6512
Dry	1/8-inch	5	4.8764	0.1137	76	42.88	<.0001	0.05	4.6499	5.1029
Wet	1/4-inch	1	4.2399	0.1137	76	37.28	<.0001	0.05	4.0134	4.4663
Wet	1/4-inch	2	4.5228	0.1137	76	39.77	<.0001	0.05	4.2963	4.7493
Wet	1/4-inch	3	4.5336	0.1137	76	39.87	<.0001	0.05	4.3071	4.7601
Wet	1/4-inch	4	4.3566	0.1137	76	38.31	<.0001	0.05	4.1301	4.5831
Wet	1/4-inch	5	4.2990	0.1137	76	37.80	<.0001	0.05	4.0725	4.5255
Wet	1/8-inch	1	4.6971	0.1137	76	41.30	<.0001	0.05	4.4706	4.9236
Wet	1/8-inch	2	4.8091	0.1137	76	42.29	<.0001	0.05	4.5826	5.0356
Wet	1/8-inch	3	4.5298	0.1137	76	39.83	<.0001	0.05	4.3034	4.7563
Wet	1/8-inch	4	4.2229	0.1137	76	37.13	<.0001	0.05	3.9964	4.4494
Wet	1/8-inch	5	4.1505	0.1137	76	36.50	<.0001	0.05	3.9240	4.3770

Figure 10: LS Means table.

LS mean comparison for interaction

The table summarizes how the response variable changes across combinations of inoculation methods, thickness, and weeks. For example, for the 1/4-inch thickness, both Dry and Wet methods show increases over time, but the Dry method typically has higher responses. Similarly, for the 1/8-inch thickness, the differences between Dry and Wet methods are smaller but still present. Confidence intervals provide assurance about the reliability of these estimates, and all results are statistically significant, confirming the trends observed.

This output provides detailed results for the interaction of Inoculation_Method and Thickness. The least squares means (LSMeans) indicate that for both Dry and Wet inoculation

methods, 1/8-inch thickness has significantly higher responses compared to 1/4-inch thickness ($p<0.0001$). Pairwise comparisons confirm significant differences, such as Dry 1/4-inch vs. 1/8-inch ($p<0.0001$) and Wet 1/4-inch vs. 1/8-inch ($p=0.0209$). Differences between inoculation methods (e.g., Dry 1/4-inch vs. Wet 1/4-inch) are also significant, with Wet showing slightly higher means. These findings highlight strong interaction effects, emphasizing that the relationship between inoculation method and response varies depending on the material thickness.

Overall, both the inoculation method (Dry or Wet) and the thickness (1/4-inch or 1/8-inch) significantly affect the outcome, and these effects depend on each other. For both methods, the 1/8-inch thickness consistently leads to higher responses compared to 1/4-inch thickness. When comparing Dry and Wet methods, Wet generally performs slightly better, especially for 1/4-inch thickness. However, the differences between Dry and Wet are more noticeable at 1/8-inch thickness, highlighting that the combination of method and thickness plays a key role. These findings suggest that to optimize results, we need to carefully consider the interaction between the method used and the thickness of the material.

Inoculatio*Thickness Least Squares Means

Inoculation_Method	Thickness	Estimate	Standard Error	DF	t Value	Pr > |t|	Alpha	Lower	Upper
Dry	1/4-inch	4.2949	0.09141	76	46.98	<.0001	0.05	4.1128	4.4770
Dry	1/8-inch	4.6997	0.09141	76	51.41	<.0001	0.05	4.5177	4.8818
Wet	1/4-inch	4.3904	0.09141	76	48.03	<.0001	0.05	4.2083	4.5724
Wet	1/8-inch	4.4819	0.09141	76	49.03	<.0001	0.05	4.2998	4.6639

Differences of Inoculatio*Thickness Least Squares Means

Inoculation_Method	Thickness	_Inoculation_Method	_Thickness	Estimate	Standard Error	DF	t Value	Pr > |t|	Alpha	Lower	Upper
Dry	1/4-inch	Dry	1/8-inch	-0.4048	0.03878	76	-10.44	<.0001	0.05	-0.4821	-0.3276
Dry	1/4-inch	Wet	1/4-inch	-0.09547	0.03878	76	-2.46	0.0161	0.05	-0.1727	-0.01824
Dry	1/4-inch	Wet	1/8-inch	-0.1870	0.03878	76	-4.82	<.0001	0.05	-0.2642	-0.1097
Dry	1/8-inch	Wet	1/4-inch	0.3094	0.03878	76	7.98	<.0001	0.05	0.2321	0.3866
Dry	1/8-inch	Wet	1/8-inch	0.2179	0.03878	76	5.62	<.0001	0.05	0.1406	0.2951
Wet	1/4-inch	Wet	1/8-inch	-0.09150	0.03878	76	-2.36	0.0209	0.05	-0.1687	-0.01427

Tests of Effect Slices for Inoculatio*Thickness Sliced By Inoculation_Method

Inoculation_Method	Num DF	Den DF	F Value	Pr > F
Dry	1	76	108.99	<.0001
Wet	1	76	5.57	0.0209

Tests of Effect Slices for Inoculatio*Thickness Sliced By Thickness

Thickness	Num DF	Den DF	F Value	Pr > F
1/4-inch	1	76	6.06	0.0161
1/8-inch	1	76	31.56	<.0001

Figure 11: fig

Inoculation Method ×Week

The interaction between inoculation method and time (Week) is evident, with Dry inoculation showing an increase in Week 5 and Wet inoculation peaking earlier. This indicates that the timing of bacterial growth depends on the inoculation method used.

Inoculation_Met*Week Least Squares Means									
Inoculation_Method	Week	Estimate	Standard Error	DF	t Value	Pr > |t|	Alpha	Lower	Upper
Dry	1	4.5002	0.1013	76	44.41	<.0001	0.05	4.2984	4.7020
Dry	2	4.4690	0.1013	76	44.10	<.0001	0.05	4.2671	4.6708
Dry	3	4.3912	0.1013	76	43.33	<.0001	0.05	4.1894	4.5930
Dry	4	4.3655	0.1013	76	43.08	<.0001	0.05	4.1637	4.5673
Dry	5	4.7607	0.1013	76	46.98	<.0001	0.05	4.5589	4.9625
Wet	1	4.4685	0.1013	76	44.10	<.0001	0.05	4.2666	4.6703
Wet	2	4.6659	0.1013	76	46.05	<.0001	0.05	4.4641	4.8677
Wet	3	4.5317	0.1013	76	44.72	<.0001	0.05	4.3299	4.7335
Wet	4	4.2898	0.1013	76	42.33	<.0001	0.05	4.0879	4.4916
Wet	5	4.2248	0.1013	76	41.69	<.0001	0.05	4.0230	4.4266

Figure 12: fig

The interaction between Inoculation_Method and Week is significant, with differences in bacterial growth patterns being particularly pronounced for Week 5, where the Dry method shows significantly higher responses than Wet. This suggests the impact of inoculation method depends on the week of measurement.

Tests of Effect Slices for Inoculation_Met*Week Sliced By Inoculation_Method				
Inoculation_Method	Num DF	Den DF	F Value	Pr > F
Dry	4	76	9.04	<.0001
Wet	4	76	14.99	<.0001

Tests of Effect Slices for Inoculation_Met*Week Sliced By Week				
Week	Num DF	Den DF	F Value	Pr > F
1	1	76	0.19	0.6649
2	1	76	7.28	0.0086
3	1	76	3.71	0.0579
4	1	76	1.08	0.3026
5	1	76	53.91	<.0001

The results show that bacterial

growth varies significantly across weeks for both thicknesses, with the effect being stronger for 1/8-inch thickness ($F=17.53, p<0.0001$) than for 1/4-inch ($F=3.19, p=0.0179$). Thickness differences are most pronounced in Weeks 1 and 2, where 1/8-inch slices show significantly higher bacterial growth compared to 1/4-inch ($p<0.0001$). By Week 5, the effect of thickness is no longer significant ($F=0.32, p=0.5723$), suggesting bacterial growth levels stabilize between the two thicknesses.

1/8-inch thickness facilitates higher bacterial growth compared to 1/4-inch across all weeks, with Week 2 showing the highest LSMean for 1/8-inch. This highlights that the combination of thinner slices and Week 2 conditions create the most favorable environment for bacterial growth.



Thickness*Week Least Squares Means

Thickness	Week	Estimate	Standard Error	DF	t Value	Pr > |t|	Alpha	Lower	Upper
1/4-inch	1	4.2195	0.1013	76	41.64	<.0001	0.05	4.0177	4.4214
1/4-inch	2	4.3300	0.1013	76	42.73	<.0001	0.05	4.1282	4.5318
1/4-inch	3	4.3601	0.1013	76	43.03	<.0001	0.05	4.1583	4.5619
1/4-inch	4	4.3315	0.1013	76	42.75	<.0001	0.05	4.1297	4.5333
1/4-inch	5	4.4721	0.1013	76	44.13	<.0001	0.05	4.2702	4.6739
1/8-inch	1	4.7491	0.1013	76	46.87	<.0001	0.05	4.5473	4.9510
1/8-inch	2	4.8049	0.1013	76	47.42	<.0001	0.05	4.6030	5.0067
1/8-inch	3	4.5628	0.1013	76	45.03	<.0001	0.05	4.3610	4.7646
1/8-inch	4	4.3238	0.1013	76	42.67	<.0001	0.05	4.1220	4.5256
1/8-inch	5	4.5135	0.1013	76	44.54	<.0001	0.05	4.3116	4.7153

Figure 13: fig

The 1/8-inch thickness shows greater variability in bacterial growth across weeks, highlighting its susceptibility. Thickness differences are most significant in Weeks 1 and 2, where thinner slices (1/8-inch) show much higher bacterial growth than thicker slices (1/4-inch). By Week 5, the effect of thickness diminishes, suggesting bacterial growth stabilizes between the two thicknesses.

Tests of Effect Slices for Thickness*Week Sliced By Thickness				
Thickness	Num DF	Den DF	F Value	Pr > F
1/4-inch	4	76	3.19	0.0179
1/8-inch	4	76	17.53	<.0001

Tests of Effect Slices for Thickness*Week Sliced By Week				
Week	Num DF	Den DF	F Value	Pr > F
1	1	76	52.64	<.0001
2	1	76	42.32	<.0001
3	1	76	7.71	0.0069
4	1	76	0.01	0.9164
5	1	76	0.32	0.5723

Figure 14: fig

Write down the model structure

Contrasts				
Label	Num DF	Den DF	F Value	Pr > F
Dry vs Wet	1	76	4.98	0.0286
1/4 vs 1/8 inches	1	76	81.92	<.0001
Dry vs Wet at 1/4 Inches	1	76	6.06	0.0161
Dry vs Wet at 1/8 Inches	1	76	31.56	<.0001
1/4 vs 1/8 inches for Dry inoculation	1	76	108.99	<.0001
1/4 vs 1/8 inches for Wet inoculation	1	76	5.57	0.0209

Figure 15: Contrasts table.

Conclusion

Future Work

References

Bobbit, Zach. 2021. "How to Interpret Negative AIC Values." Statology. 2021. https://www.statology.org/negative-aic/.

Brown, Jessica, Taylor Bannon, Arie Havelaar, Chad Carr, Barbara B. Kowalcyk, and Jason M. Scheffler. 2024. "Inactivation of Salmonella Enterica, Escherichia Coli O157:H7, and Campylobacter Jejuni in a Restructured Beef Jerky Developed for Production in Ethiopia." *Meat and Muscle Biology* 8 (April): 1–12. https://doi.org/10.22175/mmb.16091.

Lipka, Alex, and Benjamin Tyner. 2004. "Repeated Measures Covariance Structure." Statistical Consulting Service - Purdue University. 2004. https://www.stat.purdue.edu/scs/help/method/Repeated%20measures%20analysis.pdf.

USDA. n.d. "Appendix e: Selecting an Appropriate Covariance Structure," n.d., E.1–8. https://www.ars.usda.gov/ARSUserFiles/80000000/StatisticsGroupWebinars/Appendix%20E%20-%20Selecting%20a%20Covariance%20Structure.pdf.

Appendix A - R Code

Appendix B - SAS Code

Power Analysis

```
data rptm_means;
input Inoculation_Method $ Thickness $ @@;
do Week=1 to 5 by 1;
    input mu @@;
    output;
end;
datalines;
Dry 1/4 4.26 4.25 4.47 4.33 4.54
Dry 1/8 4.91 4.95 4.67 4.56 4.97
Wet 1/4 4.21 4.57 4.65 4.49 4.38
Wet 1/8 4.86 4.78 4.62 4.32 4.22
;

data rptm_design;
 set rptm_means;
 do Batches = 1 to 5; /* Creating 3 blocks (batches) */
  output;
 end;
run;

proc print data=rptm_design;
run;

/* Creating Model */

proc glimmix data=rptm_design;
    class Batches Inoculation_Method Thickness Week;
    model mu = Inoculation_Method|Thickness|Week;
    random intercept / subject=Batches;
    random Week / subject=Batches*Inoculation_Method*Thickness type=ar(1) residual;
    parms (.029)(0.017)(.028)/hold=1,2,3;
    /* Provide 3 parameters for variance components */
    lsmeans Inoculation_Method*Thickness*Week / slicediff=Week cl;
    /* Define main effect contrasts */
    contrast 'Dry vs Wet'
        Inoculation_Method 1 -1;
    contrast '1/4 vs 1/8 inches'
```

```
        Thickness 1 -1;

    /* Define interaction contrasts */
   contrast 'Dry vs Wet at 1/4 Inches'
        Inoculation_Method 1 -1 Inoculation_Method*Thickness 1 0 -1 0;
    contrast 'Dry vs Wet at 1/8 Inches'
        Inoculation_Method 1 -1 Inoculation_Method*Thickness 0 1 0 -1;
    contrast '1/4 vs 1/8 inches for Dry inoculation'
        Thickness 1 -1 Inoculation_Method*Thickness 1 -1 0 0;
    contrast '1/4 vs 1/8 inches for Wet inoculation'
        Thickness 1 -1 Inoculation_Method*Thickness 0 0 1 -1;

    ods output contrasts=f_contrast tests3=f_anova;
run;

/*Power*/
data power;
    set f_contrast f_anova;
    ncparm = numdf * fvalue;
    alpha = 0.05;
    fcrit = finv(1-alpha, numdf, dendf, 0);
    power = 1 - probf(fcrit, numdf, dendf, ncparm);
run;

proc print data=power;
run;
```

Simulation

```
/* Step 1: Define AR(1) Covariance Structure in PROC IML */
proc iml;
    n = 20;                    /* Number of subjects per treatment*/
    mean = {0 0 0 0 0};        /* Mean for each week */
    T = 5;                     /* Number of repeated measures (weeks) */
    rho = 0.2;                 /* AR(1) correlation parameter */
    sigma2 = {0.29 0.29 0.29 0.29 0.29};      /* Variance for each week */

    /* Construct AR(1) covariance matrix */
    cov = j(T, T, 0);
    do i = 1 to T;
```

```
        do j = 1 to T;
            cov[i, j] = sqrt(sigma2[i] * sigma2[j]) * rho**abs(i - j);
        end;
    end;

    /* Print covariance matrix */
    print "Covariance Matrix:", cov;

    /* Generate simulated data using the covariance matrix */
    call randseed(12349);      /* Set random seed */
    x = randnormal(n, mean, cov); /* Simulate AR(1) correlated data */
    cname = {"t1", "t2", "t3", "t4", "t5"};

    /* Print the simulated data matrix directly */
    print "Simulated Data Matrix (x):", x;
    /* Print Sample mean */
    samplemean = x[:,];
    print samplemean n;

    /* Create dataset from simulated data */
    create inputdatacb from x[colname=cname];
    append from x;
close inputdatacb;
quit;

/* Step 2: Display the Simulated Data as a SAS Table */
proc print data=inputdatacb label;
    title "Simulated Data with AR(1) Covariance Structure";
run;

/* Step 3: Define Treatment Structure and Random Effects */
data rptm_simulation;
    retain Subject 0;
    keep Inoculation_Method Thickness Week Batches Response;

    array weeks[5] t1-t5;

    /* Define mean values for each combination of factors and week */
    if _n_ = 1 then do;
        array mean_values[4,2,5] _temporary_ (
            /* Dry, 1/4 inch */
            4.26, 4.25, 4.47, 4.33, 4.54,
```

```
            /* Dry, 1/8 inch */
            4.91, 4.95, 4.67, 4.56, 4.97,
            /* Wet, 1/4 inch */
            4.21, 4.57, 4.65, 4.49, 4.38,
            /* Wet, 1/8 inch */
            4.86, 4.78, 4.62, 4.32, 4.22
        );
    end;

    /* Simulation parameters */
    sigma_batch = sqrt(0.029); /* Batch variance */
    sigma_resid = sqrt(0.017); /* Residual variance */

    /* Loop through each combination of factors */
    do Batches = 1 to 5; /* Number of batches */
        batch_effect = rand("Normal", 0, sigma_batch); /*Random batch effect*/

        do Inoculation_Method = "Dry", "Wet";
            do Thickness = "1/4-inch", "1/8-inch";
                Subject + 1;
                set inputdatacb;

                /* Generate response for each week with AR(1) structure */
                do Week = 1 to 5;
                    Mean_Value = mean_values[
                      (Inoculation_Method="Dry")*1+(Inoculation_Method="Wet")*2,
                      (Thickness="1/4-inch")*1 + (Thickness="1/8-inch")*2,
                        Week
                    ];
                    Response = Mean_Value + batch_effect + weeks[Week];
                    output;
                end;
            end;
        end;
    end;
run;

/* Step 4: Display the Simulated Data in a Structured Format */
proc print data=rptm_simulation label;
    title "Simulated Data for 2x2 Factorial Design with Repeated Measures";
run;
```

Analysis

```
proc glimmix data=data plots=residualpanel;
    class Batches Inoculation_Method Thickness Week;
    model Response = Inoculation_Method|Thickness|Week;
    random intercept / subject=Batches;
    random Week / subject=Batches*Inoculation_Method*Thickness type=ar(1) residual;
    lsmeans Inoculation_Method*Thickness*Week / adjust=tukey cl;

    /* Define main effect contrasts */
    contrast 'Dry vs Wet'
        Inoculation_Method 1 -1;
    contrast '1/4 vs 1/8 inches'
        Thickness 1 -1;

    /* Define interaction contrasts */
   contrast 'Dry vs Wet at 1/4 Inches'
        Inoculation_Method 1 -1 Inoculation_Method*Thickness 1 0 -1 0;
    contrast 'Dry vs Wet at 1/8 Inches'
        Inoculation_Method 1 -1 Inoculation_Method*Thickness 0 1 0 -1;
    contrast '1/4 vs 1/8 inches for Dry inoculation'
        Thickness 1 -1 Inoculation_Method*Thickness 1 -1 0 0;
    contrast '1/4 vs 1/8 inches for Wet inoculation'
        Thickness 1 -1 Inoculation_Method*Thickness 0 0 1 -1;

    ods output contrasts=f_contrast tests3=f_anova;
run;
```